PharmaSUG 2015 - Paper QT19

Using the power of new SGPLOT features in SAS 9.4 – Customized graphic programming made easier for clinical efficacy and exposure-response analyses

Peter Lu, Novartis Pharmaceuticals Corporation, East Hanover, New Jersey, USA
Xingji Han, Novartis Pharmaceuticals Corporation, East Hanover, New Jersey, USA
Hong Yan, Novartis Pharmaceuticals Corporation, East Hanover, New Jersey, USA

ABSTRACT
Generating customized statistical graphs for clinical efficacy and exposure-response analyses is essential in clinical study reports but this often requires advanced SAS graphic programming skills and is time-consuming. This paper explores several new features in SGPLOT, implemented in SAS® 9.4 (TS1M2), which allow us to generate customized graphs without writing lengthy code, complex macros, or having to customize the graphic template. Examples are given to demonstrate how to align an axis text-table for a Kaplan-Meier (K-M) curve, insert customized legends, overlay graphs, add unicode symbols and jitter data points without manipulating data. In the first example, two statements, XAXISTABLE and TEXT in SGPLOT, are used to align an axis text-table (e.g., number of patients at risk) and add customized legends to a K-M curve (e.g., median time-to-event, hazard ratio, time-point event-free rate estimates etc.). The second example uses a new function in SGPLOT to create overlaid graphs (e.g., a boxplot and a scatter plot) for the exposure-response analysis (drug concentration and QT change from baseline vs time). The third example uses the SYMBOLCHAR statement to add various symbols (e.g., unicode) and an option JITTER to jitter the data points easily in a scatter plot. These new SGPLOT features can be easily used by any level of SAS users from beginner to expert.

INTRODUCTION
Customized graphs are widely used in presenting clinical efficacy and exposure-response analysis results for clinical trial data. A typical example is the Kaplan-Meier (K-M) survival curve which presents time-to-event such as overall survival and progression-free survival. This figure requires customization since PROC LIFETEST only outputs a basic survival curve. There are several ways in SAS that have been used in the past to create customized K-M curves (e.g., to align the number of patients at risk with corresponding tick values at the event-time axis and to insert the descriptive and inference statistics at appropriate positions inside the graph). For instance (and prior to SAS 9.2 release), PROC LIFETEST combined with G PLOT using ANNOTATE and OVERLAY options, dynamic formatting and data step annotation. This is quite time-consuming with lengthy coding or macros. SAS Graph Template Language (GTL) in SAS 9.2 allows users to modify the template and to render the template to customize the graph, but this requires knowledge and experience in GTL.

Another example of customization is the overlaying of graphs for a comprehensive comparison of different end points and data distributions across different exposure levels. For example, to overlay a boxplot and scatter plot, programmers previously had to develop lengthy code, either embedded within a single PROC SGPLOT or combined with the graphic template language (GTL with PROC TEMPLATE and PROC SGRENDER). In the case of tied data points, it was necessary to add a small random shift for each tied data point to distinguish these equal points. In addition, symbols in SGPLOT for visualizing data points are quite limited, and many symbols defined by unicode are not available before SAS 9.4.

This paper shows how some of the new SGPLOT features in the recent release of SAS 9.4 (TS1M2) can make customized graphs easier to program. These new features are simple and easy to use and significantly improve programming efficiency.

EXAMPLE 1: ALIGNING A TEXT TABLE WITH A FIGURE X-AXIS AND INSERTING CUSTOMIZED LEGENDS FOR A K-M CURVE.

This example shows how the new statements XAXISTABLE and TEXT can be used in SGPLOT to insert the number of patients at risk and to customize the legends (Figure 1).
Using the power of SGPLOT new features in SAS 9.4 made customized graphic programming easier in clinical efficacy and exposure-response analyses, continued.

Figure 1. An example of K-M curves with number of patients at risk and summary statistical information.

The XAXISTABLE statement inserts the number of patients at risk at specific positions along the x-axis. The variable (ATRISK, number of patients at risk from PROC LIFETEST) is placed after the XAXISTABLE statement, the option X=TATRISK defines the corresponding time points, and the option class=stratum defines corresponding risk groups.

The TEXT statement inserts the customized legends and enables control of the two-dimensional (X,Y) locations in the figure. The text can be numbers or characters. In Figure 1, the location of text is specified by two required variables, x=t1 and y=s1, the content of text is provided by the third required variable text=texc1 with the option position=right. The TEXT statement also provides multiple options to better control the text appearance, e.g., to rotate the text to any angle, to split the text into multiple lines at the specified character(s) when there is not enough room to display the text normally, and to display better visual effects for the text.

In addition, it is possible to place text tables inside figures using the statement of XAXISTABLE with the option LOCATION=INSIDE. The options TITLE and TITLEATTRS can be used to give better control of the title appearance.

Code for Figure 1 (run under SAS9.4 TS1M2):
ods listing close;
ods output survivalplot=sp;
proc lifetest data=sashelp.bmt(where=(group ne 'ALL'))
    plots=survival(atrisk=0 to 1000 by 250 1000 to 2500 by 500)
    outsurv=surv timelist=(0 to 1000 by 250 1000 to 2500 by 500) reduceout;
    time T*status(0);
    strata group /test=logrank;
run;
ods listing;
data sp1;
    set sp;
    length texc1 texc2 texc3 $15;
Using the power of SGPLOT new features in SAS 9.4 made customized graphic programming easier in clinical efficacy and exposure-response analyses, continued

if stratumnum=1 then stratum='High risk';
else if stratumnum = 2 then stratum='Low risk';

if _n_=1 then do;
   t1 = 900; s1=0.95; s2=0.90; s3=0.85;
   texc1="Group"; texc2="High Risk"; texc3="Low Risk";
end;
else if _n_=2 then do;
   t1 = 1400; s1=0.95; s2=0.90; s3=0.85;
   texc1="Median (95%CI)"; texc2="183 (113-390)"; texc3="2204 (641-NE)";
end;
else if _n_=3 then do;
   t1 = 2100; s1=0.95; s2=0.90; s3=0.85;
   texc1="HR (95%CI)"; texc2="2.5 (1.2-4.8)"; texc3="Ref";
end;
else if _n_=4 then do;
   t1 = 2500; s3=0.75;
   texc3="P-value:";
end;
else if _n_=5 then do;
   t1 = 2700; s3=0.75;
   texc3="0.0002";
end;
run;

ods html gpath="/folders/myfolders/" (url=none)
   path="/folders/myfolders/" (url=none)
   file="mygraf.html" image_DPI=300 style=sasweb;

title h=9pt "Product-Limit Survival Estimates";
proc sgplot data=sp1;
   step x=time y=survival /group=stratum name='survival';
   scatter x=time y=censored /markerattrs=(symbol=plus) name='censored';
   scatter x=time y=censored /group=stratum markerattrs=(symbol=plus);
   xaxistable atrisk /class=stratum x=tatrisk;
   text x=t1 y=s1 text=texc1 /textattrs=(family="Arial" size=8 weight=bold) position=right;
   text x=t1 y=s2 text=texc2 /textattrs=(family="Arial" size=8) position=right;
   text x=t1 y=s3 text=texc3 /textattrs=(family="Arial" size=8) position=right;
   keylegend 'survival' /location=inside position=top;
   keylegend 'censored' /location=inside position=topright;
   xaxis min=0 max=2500 values=(0 to 1000 by 250 1500 to 2500 by 500) valueshint;
   yaxis offsetmin=0.02 min=0 offsetmax=0.1 max=1.0;
run;

EXAMPLE 2: OVERLAYING A SCATTER PLOT ON A BOXPLOT.

This example demonstrates that it is now possible to use VBOX plot and SCATTER plot statements together in SGPLOT to generate overlaid figures, as shown in Figures 2 and 3. In addition to scatter plots, the VBOX statement can also be combined with reference lines, other box plots, and other basic plot types including STEP, SERIES, WATERFALL, VECTOR, etc.
Using the power of SGPLOT new features in SAS 9.4 made customized graphic programming easier in clinical efficacy and exposure-response analyses, continued

Figure 2. Change of QTcF from baseline at post-dose scheduled time points.

Code for Figure 2:

```sas
title 'Overlaying a boxplot and scatter plot';
proc sgplot data=aecg;
  vbox qtcf / category=stm group=trtc fillattrs=(transparency=1) connect=mean
       connectattrs=(pattern=5) nooutliers;
  scatter y=qtcf x=stm /group=trtc groupdisplay=cluster markerattrs=(symbol=diamond size=6);
  xaxis label="Scheduled time (hr)"
       label loc=bottom;
  yaxis label="Change of QTcF (ms)" max=40;
  keylegend / title="Treatment group";
run;
```

Please note that the above code results in an error message in the log (ERROR: Attempting to overlay incompatible plot or chart type,) if used with releases prior to SAS 9.4 (TS1M1).
Using the power of SGPLOT new features in SAS 9.4 made customized graphic programming easier in clinical efficacy and exposure-response analyses, continued.

Figure 3. Change of QTcF and drug concentration at post-dose scheduled time points within a single cell.

Code for Figure 3:

```sas
title 'Change of QTcF from baseline and PK concentration';
proc sgplot data=pkecg;
    vbox qtcf / category=stm group=trtc nooutliers fillattrs=(transparency=1)
        connect=mean connectattrs=(pattern=5);
    vbox conc / category=stm group=trtc nooutliers fillattrs=(transparency=0.5)
        connect=mean connectattrs=(pattern=8) y2axis;
    scatter y=qtcf x=stm /group=trtc groupdisplay=cluster markerattrs=(symbol=diamond size=4);
    scatter y=conc x=stm /group=trtc groupdisplay=cluster
        markerattrs=(symbol=circlefilled size=4) y2axis;
    xaxis label="Scheduled time (hr)" min=-200 max=30 values=(-200 -30 -15 0 15 30);
    yaxis label="Change of QTcF (ms)" min=-200 max=30 values=(-200 -30 -15 0 15 30);
    y2axis label="Concentration (ng/mL)" min=0 max=12000 values=(0 to 8000 by 2000 12000);
    keylegend / title="Treatment group";
run;
```

Example 3: Jittering data points and adding unicode symbols.

This example shows how to add jittering and unicode for an overlaid graph using the options JITTER and the statement SYBMO CHAR, respectively, as illustrated in Figure 4.

The option JITTER can separate the tied data points and JITTER WIDTH gives the flexibility to adjust the width of jittering. The statement SYBMO CHAR defines a marker symbol using a unicode character that can be referenced in other statements. There are two required arguments, NAME=identifier and CHAR="hex-string" | keyword, where the
Using the power of SGPLOT new features in SAS 9.4 made customized graphic programming easier in clinical efficacy and exposure-response analyses, continued

“hex-string”x specifies a four-byte hexadecimal constant that can be found at http://www.unicode.org, and the keyword specifies a SAS keyword for a unicode character such as alpha and beta.

---

Figure 4. Adding jitter and unicode symbol (crossed times) for the change of QTcF at time points.

Code for Figure 4:

```sas
title 'Overlaying a boxplot and scatter plot with jittering effects';
proc sgplot data=aecg(where=(stm<24));
    vbox qtcf / category=stm group=trtc fillattrs=(transparency=1) nooutliers nomean;
    symbolchar name=mys char="2297" /textattrs=(family="arial" weight=bold);
    scatter y=qtcf x=stm /group=trtc groupdisplay=cluster
        markerattrs=(symbol=mys size=12) jitter;
    xaxis label="Scheduled time (hr)";
    yaxis label="Change of QTcF (ms)" max=40;
    keylegend / title="Treatment group";
run;
```

CONCLUSION

This paper demonstrates the power of new SGPLOT features in SAS 9.4 for generating customized graphs. Three examples show clearly that these updates are straight-forward, easy to use, and can improve statistical programming efficiency in the pharmaceutical industry.

ACKNOWLEDGMENTS

We thank our colleague Danny Tuckwell for his review and editorial comments in the preparation of this paper.
Using the power of SGPLOT new features in SAS 9.4 made customized graphic programming easier in clinical efficacy and exposure-response analyses, continued

REFERENCES

Dong, F., Customizing Survival Plot Using ODS Graphics Template Language, PharmaSUG2013 (Paper DG05)
Harm, V., Grouped Jittered Box Plots in SAS 9.2 and SAS 9.3, PHUSE 2012, paper CS03.
Yu, X., Dances with Box Plot, PharmaSUG 2013, paper DG02.
Minard, C.G., Overlaying Scatter Plots on Box Plots in the Presence of Ties.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the authors at:

Name: Peter Lu
Enterprise: Novartis Pharmaceuticals Corporation,
Address: One health plaza, East Hanover, New Jersey 07936-1080
E-mail: peter.lu@novartis.com

Name: Xingji Han
Enterprise: Novartis Pharmaceuticals Corporation,
Address: One health plaza, East Hanover, New Jersey 07936-1080
E-mail: xingji.han@novartis.com

Name: Hong Yan
Enterprise: Novartis Pharmaceuticals Corporation,
Address: One health plaza, East Hanover, New Jersey 07936-1080
E-mail: hong.yan@novartis.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.
Other brand and product names are trademarks of their respective companies.